Math 173
Calculus I

Dr. Constant J. Goutziers
Department of Mathematics, Computer Science and Statistics
goutzicj@oneonta.edu

Lesson 3

One-to-one Functions and Inverse Functions

Initializations

```
> restart;
  with(plots):
>
```

3.1 One-to-one Functions

A function f with domain A is called one-to-one if no two elements of A have the same image; that is $f(x_1) = f(x_2)$ implies that $x_1 = x_2$. This property means that a horizontal line can intersect the graph of a one-to-one function at most once, which is known as the horizontal line test.

Examples

Example 3.1.1

Use the horizontal line test to show that the function $f(x) = x^2 + 3x + 7$ is not one-to-one.

Solution
We make a sketch.

```
> f:=x->x^2+3*x+7;
  plot(f(x), x=-4..3);
```

$$f := x \rightarrow x^2 + 3x + 7$$




```
>
```

Clearly, there are many horizontal lines which intersect with this graph more than once. The line $y = 9$ is an example as can be seen in the next graph.

```
> plot({f(x), 9}, x=-4..3);
```



The function f is therefore not one-to-one.

Example 3.1.2

Use the definition of a one-to-one function to show that $f(x) = \frac{3x+2}{x-7}$, $x \neq 7$ is one-to-one.

Solution

We will show that $f(x_1) = f(x_2)$ implies that $x_1 = x_2$.

```
> f:=x->(3*x+2)/(x-7);
```

$$f := x \rightarrow \frac{3x+2}{x-7} \quad \textbf{(2.1.2.1)}$$

We assume that x_1 and x_2 do not equal 7 and solve the equation $f(x_1) = f(x_2)$ for x_1.

```
> eq:=f(x[1])=f(x[2]);
```

$$eq := \frac{3x_1+2}{x_1-7} = \frac{3x_2+2}{x_2-7} \quad \textbf{(2.1.2.2)}$$

```
> test:=solve(eq, {x[1]});
```

$$test := \{x_1 = x_2\} \quad \textbf{(2.1.2.3)}$$

Since $x_1 = x_2$ is the only solution, we conclude that the function f is one-to-one.

```
>
```

3.2 Inverse Functions

A one-to-one function f with domain A and range B has an inverse function f^{-1} defined by

$$f^{-1}(y) = x \quad \text{if and only if} \quad f(x) = y$$

The inverse function f^{-1} has domain B and range A.

Examples

Example 3.2.1

Compute the inverse function of $f(x) = \frac{5x}{4-3x}$**,** $0 \le x \le 1$**.**

Solution
First we plot the function *f*.

```
> f:=x->5*x/(4-3*x);
  plot(f(x), x=0..1);
```

$$f := x \to \frac{5x}{4-3x}$$




```
>
```

Clearly, the function passes the horizontal line test, so it has an inverse. Observe that the domain *A* of *f* is the interval $[0, 1]$, and the range *B* of *f* is the interval $[f(0), f(1)] = [0, 5]$. To find f^{-1} we solve the equation $y = f(x)$ for *x*.

```
> eq:=y=f(x);
```

$$eq := y = \frac{5x}{4-3x} \qquad \textbf{(3.1.1.1)}$$

```
> finv_y:=solve(eq, x);
```

$$finv\_y := \frac{4y}{3y+5} \qquad \textbf{(3.1.1.2)}$$

By interchanging *x* and *y*, we obtain the formula for $f^{-1}(x)$.

```
> finv_x:=subs(y=x, finv_y);
```

$$finv\_x := \frac{4x}{3x+5} \qquad \textbf{(3.1.1.3)}$$

If so desired, this Maple expression can be converted into a Maple function by using the **unapply** command.

```
> finv:=unapply(finv_x, x);
```

$$finv := x \to \frac{4x}{3x+5} \qquad \textbf{(3.1.1.4)}$$

The domain of f^{-1} is the range *B* of *f*, so

$$f^{-1}(x) = \frac{4x}{3x+5},\ 0 \le x \le 5$$

Example 3.2.2

Consider the function $f(x) = \sqrt{x-5}, 5 \le x \le 8$

i) Show that *f* is one-to-one.

ii) Find the inverse of *f*.

iii) Sketch f and its inverse in one picture

Solution

i) Show that *f* is one-to-one.

We use the horizontal line test.

```
> f:=x->sqrt(x-5);
  plot(f(x), x=5..8);
```

$$f := x \to \sqrt{x-5}$$




```
>
```

No horizontal line intersects with the curve more than once, so the function is one-to-one.

ii) Find the inverse of *f*.

The inverse of *f* can be found in the usual way.

```
> eq:=y=f(x);
```

$$eq := y = \sqrt{x-5} \quad (3.1.2.1)$$

```
> finv_y:=solve(eq, x);
```

$$finv\_y := 5 + y^2 \quad (3.1.2.2)$$

```
> finv_x:=subs(y=x, finv_y);
```

$$finv\_x := 5 + x^2 \quad (3.1.2.3)$$

```
> finv:=unapply(finv_x, x);
```

$$finv := x \to 5 + x^2 \quad (3.1.2.4)$$

```
>
```

The domain of f^{-1} equals the range of *f*, therefore

$$f^{-1}(x) = 5 + x^2,\ 0 \le x \le \sqrt{3}$$

iii) Sketch f and its inverse in one picture.

When it comes to plotting the functions f and f^{-1} we have to keep in mind that they have different domains. Clearly the function f has [5, 8] as its domain, but the domain of f^{-1} equals $\left[0, \sqrt{3}\right]$. In order to handle these different domains we create two separate graphic images, one for f and the other for f^{-1}.

```
> p1:=plot(f(x), x=5..8):
```

```
> p2:=plot(finv(x), x=0..sqrt(3), color=blue):
```

Observe that these statements end with a colon rather than a semicolon. This prevents premature display of the output. We now use the **display** command to simultaneously show the graphs of f and f^{-1}. If we choose equal scaling on the coordinate axes, the two graphs are each others mirror image in the line $y = x$. Equal scaling is obtained by using the **scaling=constrained** option in the plot command

```
> display([p1, p2], scaling=constrained);
```




```
>
```

This mirror image property of graphs of inverse functions allows us to quickly sketch the graph of a one-to-one function and its inverse without actually computing that inverse. To do this we make use of a parametric plot, as demonstrated in the next example.

Example 3.2.3

Sketch in one picture the graph of the one-to-one function $f(x) = x^3 + 2$ and its inverse.

Solution

We use Maple's parametric plot routine and to ensure equal scaling on the coordinate axes we again include the **scaling=constrained** option.

```
> f:=x->x^3+2;
```

$$f := x \rightarrow x^3 + 2 \tag{3.1.3.1}$$

```
> plot({[t, f(t), t=-1.3..1], [f(t), t, t=-1.3..1]},
  scaling=constrained);
```



The line $y = x$ can be added to the picture, emphasizing the mirror property

```
> plot([[t, f(t), t=-1.3..1], [f(t), t, t=-1.3..1], [t, t,
  t=-1..3]], color=[red, green, blue], scaling=constrained)
  ;
```




```
>
```